
ABSTRACT 
Creating dynamic, data driven programs is a very powerful tool.  Often we can use the metadata or the project data itself to help write our 
programs.  This paper will introduce some of the concepts related to writing SAS® programs which will generate pieces of SAS code in a data 
driven manner.   
 
Parts of the SAS language used in this discussion include the following: 
 
1. macro call routines (call execute, call symput) 
2. PROC SQL 
3. SAS Dictionary Tables 
4. Stored Processes 
 
Programming techniques used in this discussion include the following: 
 
1. using sequential macro variables  with macro do loops 
2. generating SAS code, writing it to an external file, and %including it at a later time 
3. generating and executing SAS code using call execute 
4. checking if a data set exists 
5. checking if a variable exists in a data set 
6. checking if a data set is empty 
7. using subqueries for multiple data set  subsetting 
 

INTRODUCTION 
A major problem that can occur in project code is hard coding project specific facts into a program.  A good example is a program used across 
multiple protocols in a clinical trial.  For instance, we may know that all protocols except for 3 and 5 already have the variable age calculated in 
the demographics module.  This may tempt us to have something like the following in our program. 
 
%if &prot.=3 or &prot=5 %then 
%do; 
%*  calculate the age variable.; 
%end; 
 
This works fine but what happens if more protocols come up at a later time which also lack the age variable.  The code will have to be 
changed when this happens.  It may be beneficial to programmatically check to see if the variable age exists in the demographics module.  
Based on that data driven check, the age variable can be calculated if necessary. 
 
Using the parts of the SAS language and programming techniques discussed in this paper can result in low maintenance, reusable programs.  
They can also sometimes save a lot of typing! 
 
This paper is intended for those with previous exposure to macro and SQL.  Please be advised that some of these SAS tools are not available 
prior to SAS version 6.12. 
 

PARTS OF THE SAS LANGUAGE 
First we will review the parts of the SAS language utilized in this paper. 
 

1.  CALL SYMPUT 
Call symput is a data step statement used to create a macro variable from a value or data set variable.  Following is an example… 
 
data _null_; 
   call symput(‘macvar’,’macro variable value’); 
run; 
 
The first argument is the name of the macro variable.  The second argument is the macro variable value.  This can also be a data set variable. 
 



CALL EXECUTE 
Call execute is also a data step statement which is used to execute whatever is passed into it.  An example follows… 
 
data _null_; 
   call execute(‘proc options; run;’); 
run; 
 
There is only one argument in a call execute. 
 
Caution:  There is a version 7 bug which compresses out spaces when combining multiple call executes.  For example, the following code 
runs in version 6.12 but not in version 7. 
 
data _null_; 
   call execute(‘proc ’); 
   call execute(‘options; run;’); 
run; 
 
Version 7 ends up trying to execute the code 
 
procoptions; run; instead of proc options; run; 
 
This was fixed in Version 8. 

2.  PROC SQL 
PROC SQL will be used in this paper to access the dictionary tables as well as create macro variables similar to how we can use call symput.   
 
An example follows which places the data set label from  SASHELP.RETAIL into a macro variable called dslabel; 
 
proc sql noprint; 
    select memlabel into :dslabel 
    from dictionary.tables 
    where libname=’SASHELP’ and memname=’RETAIL’; 
quit; 
 
Following is an example creating sequential macro variables from multiple records.  This loads variable names from SASHELP.RETAIL into 
macro variables var1, var2, …, varn.  Even though room is left for 9,999 sequential macro variables only the necessary ones are created. 
 
proc sql noprint; 
    select name into :var1 - :var9999 
    from dictionary.columns 
    where libname=’SASHELP’ and memname=’RETAIL’; 
quit; 
 
Following is an example creating one concatenated macro variable from multiple records.  This loads variable names from SASHELP.RETAIL 
into macro variable varlist separated by whatever we specify.  This provides us with a handy list of variables to include in a keep= or drop= 
data set option later in the program. 
 
proc sql noprint; 
    select name into :varlist separated by ‘ ‘ 
    from dictionary.columns 
    where libname=’SASHELP’ and memname=’RETAIL’; 
quit; 
If you are interested in building up a macro variable of quoted values try the following 
 
separated by ‘ ”,” ’ 

3.  DICTIONARY TABLES 
SAS dictionary tables contain the data behind the scenes in SAS a.k.a. the metadata.  The dictionary tables are only directly accessible from 
PROC SQL.  There is also a corresponding set of data step views but it is much less efficient to use these due to where statements not getting 
pushed back to the dictionary tables.  Following is a list of dictionary tables and their contents.  There were not any new tables added in V8.  
New columns added in V8 are bolded and new columns added in V9 are shaded.  A list of tables that are new in V9 follow the list of tables 
available in V9. 
 
DICTIONARY.CATALOGS 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   OBJNAME char(8) label='Object Name', 
   OBJTYPE char(8) label='Object Type', 
   OBJDESC char(40) label='Object Description', 
   MODIFIED char(8) label='Date Modified', 
   CREATED num format=DATETIME informat=DATETIME    
                             label='Date Created', 
   ALIAS char(8) label='Object Alias' 

 



   LEVEL num label='Library Concatenation Level' 
 
DICTIONARY.COLUMNS 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   NAME char(8) label='Column Name', 
   TYPE char(4) label='Column Type', 
   LENGTH num label='Column Length', 
   NPOS num label='Column Position', 
   VARNUM num label='Column Number in Table', 
   LABEL char(40) label='Column Label', 
   FORMAT char(16) label='Column Format', 
   INFORMAT char(16) label='Column Informat', 
   IDXUSAGE char(9) label='Column Index Type' 
   SORTEDBY num label='Order in Key Sequence', 
   XTYPE char(12) label='Extended Type', 
   NOTNULL char(3) label='Not NULL?', 
   PRECISION num label='Precision', 
   SCALE num label='Scale' 
 
DICTIONARY.EXTFILES 
   FILEREF char(8) label='Fileref', 
   XPATH char(80) label='Path Name', 
   XENGINE char(8) label='Engine Name' 
 
DICTIONARY.INDEXES 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   NAME char(8) label='Column Name', 
   IDXUSAGE char(9) label='Column Index Type', 
   INDXNAME char(8) label='Index Name', 
   INDXPOS num label='Pos. of Column in Concatenated Key', 
   NOMISS char(3) label='Nomiss Option', 
   UNIQUE char(3) label='Unique Option' 
 
DICTIONARY.MEMBERS 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   ENGINE char(8) label='Engine Name', 
   INDEX char(8) label='Indexes', 
   PATH char(80) label='Path Name' 
 
DICTIONARY.OPTIONS 
   OPTNAME char(16) label='Session Option Name', 
   SETTING char(200) label='Session Option Setting', 
   OPTDESC char(80) label='Option Description' 
   LEVEL char(8) label='Option Location' 
   GROUP char(32) label='Option Group' 
 
DICTIONARY.TABLES 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   MEMLABEL char(40) label='Dataset Label', 
   TYPEMEM char(8) label='Dataset Type', 
   CRDATE num format=DATETIME label='Date Created', 
   MODATE num format=DATETIME label='Date Modified', 
   NOBS num label='Number of Observations', 
   OBSLEN num label='Observation Length', 
   NVAR num label='Number of Variables', 
   PROTECT char(3) label='Type of Password Protection', 
   COMPRESS char(8) label='Compression Routine', 
   ENCRYPT char(8) label='Encryption', 
   NPAGE num label='Number of Pages', 
   FILESIZE num label='Size of File',   
   PCOMPRESS num label='Percent Compression',    
   REUSE char(3) label='Reuse Space', 
   BUFSIZE num label='Bufsize', 
   DELOBS num label='Number of Deleted Observations', 
   NLOBS num label='Number of Logical Observations', 
   MAXVAR num label='Longest variable name', 

 



   MAXLABEL num label='Longest label', 
   MAXGEN num label='Maximum number of generations', 
   GEN num label='Generation number', 
   ATTR char(3) label='Dataset Attributes',   
   INDXTYPE char(9) label='Type of Indexes' 
   DATAREP char(32) label='Data Representation', 
   SORTNAME char(8) label='Name of Collating Sequence', 
   SORTTYPE char(4) label='Sorting Type', 
   SORTCHAR char(8) label='Charset Sorted By' 
   REQVECTOR char(24) format=$HEX informat=$HEX  
                                         label='Requirements Vector' 
   DATAREPNAME char(170)  
                                         label='Data Representation Name' 
   ENCODING char(256) label='Data Encoding' 
 
DICTIONARY.VIEWS 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(8) label='Member Name', 
   MEMTYPE char(8) label='Member Type', 
   ENGINE char(8) label='Engine Name' 
 
DICTIONARY.TITLES 
   TYPE char(1) label='Title Location', 
   NUMBER num label='Title Number', 
   TEXT char(200) label='Title Text' 
 
DICTIONARY.MACROS 
   SCOPE char(9) label='Macro Scope', 
   NAME char(8) label='Macro Variable Name', 
   OFFSET num label='Offset into Macro Variable', 
   VALUE char(200) label='Macro Variable Value' 
 
 
New Tables in V9.0! 
DICTIONARY.CONSTRAINT_COLUMN_USAGE 
   TABLE_CATALOG char(8) label='Libname', 
   TABLE_SCHEMA char(8) label='Table Schema', 
   TABLE_NAME char(32) label='Table', 
   COLUMN_NAME char(32) label='Column', 
   CONSTRAINT_CATALOG char(32) 
                 label='Constraint Catalog', 
   CONSTRAINT_SCHEMA char(8)  
                 label='Constraint Schema', 
   CONSTRAINT_NAME char(32)  
                 label='Constraint Name' 
 
DICTIONARY.DICTIONARIES 
   MEMNAME char(32) label='Member Name', 
   MEMLABEL char(256) label='Dataset Label', 
   NAME char(32) label='Column Name', 
   TYPE char(4) label='Column Type', 
   LENGTH num label='Column Length', 
   NPOS num label='Column Position', 
   VARNUM num label='Column Number in Table', 
   LABEL char(256) label='Column Label', 
   FORMAT char(49) label='Column Format', 
   INFORMAT char(49) label='Column Informat' 
 
DICTIONARY.ENGINES 
   ENGINE char(8) label='Engine Name', 
   ALIAS char(8) label='Alias', 
   DESCRIPTION char(40) label='Description', 
   PREFERRED char(3) label='Preferred?', 
   PROPERTIES char(1024) label='Engine Dialog Properties' 
 
DICTIONARY.FORMATS 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(32) label='Member Name', 
   PATH char(1024) label='Path Name', 
   OBJNAME char(32) label='Object Name', 
   FMTNAME char(32) label='Format Name', 
   FMTTYPE char(1) label='Format Type', 
   SOURCE char(1) label='Format Source', 

 



   MINW num label='Minimum Width', 
   MIND num label='Minimum Decimal Width', 
   MAXYYnum label='Maximum Width', 
   MAXD num label='Maximum Decimal Width', 
   DEFW num label='Default Width', 
   DEFD num label='Default Decimal Width' 
 
DICTIONARY.GOPTIONS 
   OPTNAME char(32) label='Option Name', 
   SETTING char(1024) label='Option Setting', 
   OPTDESC char(160) label='Option Description', 
   LEVEL char(8) label='Option Location', 
   GROUP char(32) label='Option Group' 
 
DICTIONARY.LIBNAMES 
   LIBNAME char(8) label='Library Name', 
   ENGINE char(8) label='Engine Name', 
   PATH char(1024) label='Path Name', 
   LEVEL num label='Library Concatenation Level', 
   FILEFORMAT char(8) label='Default File Format', 
   READONLY char(3) label='Read-only?', 
   SEQUENTIAL char(3) label='Sequential?', 
   SYSDESC char(1024) label='System Information Description', 
   SYSNAME char(1024) label='System Information Name', 
   SYSVALUE char(1024) label='System Information Value' 
 
DICTIONARY.REMEMBER 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(32) label='Member Name', 
   OFFSET num label='Offset into Text Remembered', 
   RTEXT char(200) label='Text Remembered', 
   PW char(8) label='Password' 
 
DICTIONARY.STYLES 
   LIBNAME char(8) label='Library Name', 
   MEMNAME char(32) label='Member Name', 
   STYLE char(32) label='Style Name', 
   CRDATE num format=DATETIME informat=DATETIME  
             label='Date Created' 
 
DICTIONARY.CONSTRAINT_TABLE_USAGE 
   TABLE_CATALOG char(8) label='Libname', 
   TABLE_SCHEMA char(8) label='Table Schema', 
   TABLE_NAME char(32) label='Table', 
   CONSTRAINT_CATALOG char(32) label='Constraint Catalog', 
   CONSTRAINT_SCHEMA char(8) label='Constraint Schema', 
   CONSTRAINT_NAME char(32) label='Constraint Name' 
 
DICTIONARY.REFERENTIAL_CONSTRAINTS 
   CONSTRAINT_CATALOG char(32) 
             label='Constraint Catalog', 
   CONSTRAINT_SCHEMA char(8)  
             label='Constraint Schema', 
   CONSTRAINT_NAME char(32) label='Constraint Name', 
   UNIQUE_CONSTRAINT_CATALOG char(32) 
             label='Unique Constraint Catalog', 
   UNIQUE_CONSTRAINT_SCHEMA char(8) 
             label='Unique Constraint Schema', 
   UNIQUE_CONSTRAINT_NAME char(32) 
             label='Unique Constraint Name', 
   MATCH_OPTION char(8) label='Match Option', 
   UPDATE_RULE char(12) label='Update Rule', 
   DELETE_RULE char(12) label='Delete Rule' 
 
DICTIONARY.TABLE_CONSTRAINTS 
   TABLE_CATALOG char(8) label='Libname', 
   TABLE_SCHEMA char(8) label='Table Schema', 
   TABLE_NAME char(32) label='Table', 
   CONSTRAINT_CATALOG char(32)  
             label='Constraint Catalog', 
   CONSTRAINT_SCHEMA char(8)  
             label='Constraint Schema', 
   CONSTRAINT_NAME char(32) label='Constraint Name', 
   CONSTRAINT_TYPE char(8) label='Constraint Type', 

 



   IS_DEFERRABLE char(1) label='Is Deferred?', 
   INITIALLY_DEFERRED char(1) label='Initially Deferred?' 
 
DICTIONARY.CHECK_CONSTRAINTS 
   CONSTRAINT_CATALOG char(32) label='Constraint Catalog', 
   CONSTRAINT_SCHEMA char(8) label='Constraint Schema', 
   CONSTRAINT_NAME char(32) label='Constraint Name',    
 

4.  STORED PROCESSES WITHIN THE SAS MICROSOFT OFFICE PLUG-IN 
Stored processes are SAS programs stored in a central location and registered with the Management Console.  These stored processes can 
be utilized by the Microsoft Office Plug-In as well as other BI tools within the new SAS 9 technology packages. 
 
Now that we have discussed the major SAS tools of interest in this paper, let’s talk about using these components to write code that writes 
code. 
 

PROGRAMMING TECHNIQUES 
Following is one problem solved using three different programming techniques.  
 
Many times we have access to files,which contain variable labels.  Instead of typing all of these in or doing some awkward cutting and pasting.  
Consider the following data set called work.vardata containing the variables var and varlabel.  A data set called demo contains the variables 
var1, var2, and var3 which we want to apply these variable labels to. 
 
 Var Varlabel 
Var1 Patient # 
Var2 Date of Visit 
Var3 Date of Birth 
 
There are various ways that we can generate the label statement programmatically.  Even though there are only three variables here, there 
are times that we are presented with hundreds or thousands of variables, which need labels generated and we do not want to type them in!  
What we want to end up with is the following label statement inside of a SAS data step or procedure. 
 
Label var1=’Patient #’  
          var2=’Date of Visit’ 
          var3=’Date of Birth’; 
 

1.  USING SEQUENTIAL MACRO VARIABLES WITH MACRO DO LOOPS 
proc sql noprint; 
   select var into :VAR1-:VAR9999 
   from vardata; 
 
   select varlabel into :VL1-:VL9999 
   from vardata; 
 
   select count(*) into :NUMVL 
   from vardata;  quit;  
 
%macro runit; 
 
proc datasets lib=work; 
    modify demo; 
    label 
             %do a=1 %to &NUMVL.; 
                &&VAR&a=”&&VL&a.” 
             %end; 
     ; 
quit; 
 
%mend runit; 
 
Note:  macro do loops must be inside of a macro definition.   
 
The double ampersand reference causes the macro reference to be scanned twice.  For instance, when a=1 &&var&a is resolved in the 
following way.   
 
First scan: when two ampersands are encountered, they are resolved into one ampersand.  The &a resolves into 1 leaving us with &var1. 
 
Second scan: &var1 resolves as it normally would.   
 

2.  GENERATING SAS CODE, WRITING IT TO AN EXTERNAL FILE, AND %INCLUDING IT AT A LATER TIME 
data _null_; 

 



   file ‘tempprog.sas’; 
   put @1 ‘proc datasets lib=work;’; 
   put @4 ‘modify demo;’; 
   put @4 ‘label ‘; 
   do until(last); 
       set vardata end=last; 
        put @7 var ’ =” ’ varlabel ‘ ” ’; 
   end; 
   put @4 ‘;’; 
   put @1 ‘quit;’; 
   stop; 
run; 
 
%include ‘tempprog.sas’; 
 

3.  GENERATING AND EXECUTING SAS CODE USING CALL EXECUTE 
data _null_; 
   call execute(‘proc datasets lib=work;’); 
   call execute(‘modify demo;’); 
   call execute(‘label ‘); 
   do until(last); 
       set vardata end=last; 
       call execute(var||’=”’||varlabel||‘”’); 
   end; 
   call execute(‘;’); 
   call execute(‘quit;’); 
   stop; 
run; 
 
Another handy thing to do programmatically is to check the status of a data set. 
 

4.  CHECKING TO SEE IF A DATA SET EXISTS 
proc sql noprint; 
   select left(put(count(*),8.)) into :exist 
   from dictionary.members 
   where libname='WORK' and 
             memname='JUNK'; 
quit; 
 
If the data set work.junk exists, the macro variable exist will be equal to 1; otherwise it will be equal to 0. 
 

5.  CHECKING TO SEE IF A VARIABLE EXISTS IN A DATA SET 
proc sql noprint; 
   select left(put(count(*),8.)) into :exist 
   from dictionary.columns 
   where libname='WORK' and 
             memname='JUNK' and  
             name=’DOG’; 
quit; 
 
If the variable dog exists in the data set work.junk, the macro variable exist will be equal to 1; otherwise it will be equal to 0. 
 

6.  CHECKING TO SEE IF A DATA SET IS EMPTY 
proc sql noprint; 
   select left(put(nobs,8.)) into :nobs 
   from dictionary.tables 
   where libname='WORK' and 
             memname='JUNK' ; 
quit; 
 

 



7.  USING SUBQUERIES FOR CROSS DATA SET SUBSETTING 
proc sql; 
   create table final as 
   select * 
   from prot008.demo 
   where patient not in (select distinct patient 
                                   from prot008.vitals);    
quit; 
 
The data set final only has records from demo for patients who do not appear in the vitals data set. 

CONCLUSION 
As one can imagine, there are many opportunities to apply these techniques in our day to day programming.  Whenever developing SAS code 
to be used over again or as part of a larger system of SAS programs, we should try to avoid hard coding information into our programs.  This 
paper only scratches the surface in providing examples of data driven programming.  Try to apply these ideas to your programming problems. 
 
For even more ideas in data driven programming please see the paper “%SYSFUNC – The Brave New Macro World” by Chris Yindra.  This 
paper begins on page 259 of the SUGI 23 Proceedings. 

